

This free back issue of
THE QUICK ANSWER is provided courtesy of...

Inside
Sesame.com

The Monthly Newsletter for Sesame Database Manager
<http://www.insidesesame.com>
[Read a Free Issue of Inside Sesame!](#)

AND

L A N T I C A

S O F T W A R E L L C .

Makers of [Sesame Database Manager](#)
Compatible with Symantec Q&A™
<http://www.lantica.com>

Versatile Merge-Faxing

TOM MARCELLUS



Running Q&A for DOS in Windows gives you some rather powerful fax-merging options.



FINDING a way to simplify my faxing turned into an interesting adventure. I store all my contact data in Q&A 5.0, which I run in Windows 95, and I use Symantec's WinFax Pro to fax right from my PC. My problem was how to merge the Q&A data with WinFax faxes.

In an ordinary merge-document, you have merge fields for the name, address, and so forth. You open the document in Write, type whatever text you need, and tell Q&A which record to merge during printing.

That's the kind of faxing simplicity I was after. But I also wanted an attractive fax with fonts and graphics I could send via WinFax. Q&A Write for DOS couldn't fill the bill, so I warmed up to Q&A Write for Windows.

Here's what I did. In Write for Windows, I created a merge document (see Figure 1) using the WinFax printer driver. This way, when I print, the data from the selected Q&A 5.0 record merges with it, and WinFax sends it as a fax. I can send great-looking personalized faxes with a few mouse clicks, and even include optional *fax attachments*.

Sending a merge-fax could hardly be easier

If you want to merge your Q&A data with faxes, this solution makes it a breeze. Once you've set things up, here's all it'll take to send a fax:

1. From the client's Q&A for DOS record, you'll click on the **Send Fax** button to switch to your Write for Windows fax-merge document.
2. If your fax message isn't already in the document, you'll type it, then click on the Print Merge icon.
3. At the Retrieve Spec, you'll select the automatically generated *Fax1* saved Spec, (See Figure 2), and WinFax's *Send* dialog will appear.
4. You'll click on the recipient's name, click on the Send button, and WinFax will transmit the fax. Couldn't be easier.




Figure 1. Click on the **Send Fax** button in the client's Q&A 5.0 DOS record (1), and your Write for Windows fax-merge document appears (2), ready to fax to that client (3). Along with the fax you can send fax attachments, such as price lists and announcements.

The Quick Answer™

The Independent Guide to Q&A Expertise

July 1997

Volume 8, Number 7

- 1 Versatile Merge-Faxing
Tom Marcellus
- 2 *Editorial:* The Amazing Fax-O-Matic
- 2 *Tip:* When Auto-Recalc Won't
Erika Yoxall
- 6 Macros—The Missing Link
Erika Yoxall
- 8 Selection Lists and Line-Item Databases
Jeff Nitka
- 11 @Help
Edited by T.J. Shufflin
- 12 *Tip:* Privatize Files
Tom Marcellus

MARBLE PUBLICATIONS

Step 1—Make your Q&A 5.0 database fax-ready

You can use any address database. I'll illustrate with a simple Clients file that includes fields for the contact (person), company, and fax number. Your fields and field names might differ—you might have first and last name fields, and separate fields for the fax area code and fax number. (See the sidebar, "Fax Area Codes.") If so, just modify the instructions accordingly.

Continues on page 3

The Amazing Fax-O-Matic

TOM MARCELLUS

DO you fax a lot? I do. If you run Q&A 5.0 in Windows 95, you can fax-merge your Q&A data with the same ease as printing a merge letter to a contact. I designed my fax-merge template in Q&A Write for Windows so I could include WYSIWYG fonts and graphics, use WinFax Pro as my faxing software, and merge my Q&A for DOS data. (I could use any Windows word processor, but then I'd have to export the merge data to a supported file format.)

To send a fax, I simply pull up the contact's record, and click on the Send Fax button to display the fax template. The merge fields are automatically filled. So, I type the message (or use a pre-composed fax message with or without additional merge fields), and with a few more mouse clicks my fax is on its way. I'll show you how to do it.

Who's afraid of macros?—even seasoned Q&A users, as **Erika Yoxall** found out. She was so flabbergasted by this macrophobia that she decided to go back to basics with a macro tutorial series beginning this month.

Speaking of automating Q&A, some types of databases just cry out for ways to simplify data entry. Sales orders, invoices, purchase orders and similar line-item databases are good examples. **Jeff Nitka** shows you a programming technique that lets you fill out records and choose options almost exclusively from pop-up lists. And his pick-lists aren't static—they present different options depending on conditions in the record.

If you've experienced the loss of all fonts when printing font-enhanced reports in Q&A 5.0, this month's @Help column describes the nature of the bug and ways to work around it.

Big Q&A Event Set for October—See page 7 for info on the upcoming National Q&A User Group's 7th Annual Bash and *The Quick Answer's* Q&A Masters Seminar

The Quick Answer™

The Independent Guide to Q&A Expertise

Editor Tom Marcellus
Publisher Michael Bell

The Quick Answer (ISSN 1052-3820) is published monthly (12 times per year) by Marble Publications, Inc., 9717 Delamere Ct., Rockville, MD 20850.

Cost of domestic subscriptions: 12 issues, \$79; 24 issues, \$142. Outside the U.S.: 12 issues, \$99; 24 issues, \$172. Single copy price: \$10; outside the U.S., \$12.50. All funds must be in U.S. currency. Back issues are available upon request, for the same price as a single copy.

Periodicals postage paid at Rockville, MD. POSTMASTER: Send address changes to **The Quick Answer, PO Box 9034, Gaithersburg, MD 20898-9034.**

Copyright © 1997 by Marble Publications, Inc. All rights reserved. No part of this periodical may be used or reproduced in any fashion (except in the case of brief quotations embodied in articles and reviews) without the prior written consent of Marble Publications, Inc.

Address editorial correspondence, @HELP questions, or requests for special permission to: Marble Publications, Inc., The Quick Answer, PO Box 9034, Gaithersburg, MD 20898-9034. Phone 800-780-5474 or 301-424-1658. Fax 301-424-1658. CompuServe 73370.1575.

For Q&A technical support, call Symantec 503-465-8600.

Q&A is a trademark of Symantec Corp. Other brand and product names are trademarks or registered trademarks of their respective holders.

This publication is intended as a general guide. It covers a highly technical and complex subject and should not be used for making decisions concerning specific products or applications. This publication is sold as is, without warranty of any kind, either express or implied, including but not limited to implied warranties for the publication, quality, performance, merchantability, or fitness for any particular purpose. Marble Publications, Inc., shall not be liable to the purchaser or any other person or entity with respect to any liability, loss, or damage caused or alleged to be caused directly or indirectly by this publication. Articles published in *The Quick Answer* do not necessarily reflect the viewpoint of Marble Publications, Inc.



When Auto-Recalc Won't



When you set a database's Calc mode to Automatic (Shift-F8 in Add Data or Search/Update mode), Q&A will recalculate the form when you exit a field after changing its contents. Most of the time, that is. Data changes involving non-typed information won't trigger the recalc. For example, suppose you have two fields programmed this way:

```
<#1: If #1 = "" Then #1 = @Userselect("1,2,3,4,5")
#2: #2 = #1 * 5
```

Here, the @Userselect must execute on-field-entry because you want the list to display when the user moves to the field. However, when you make a selection from the list and exit the field, Q&A won't recalculate field #2 until you change another

field or press F8 to manually recalc the record.

For these cases, I use an @Macro("Calculate") command to force the recalculation. The Calculate macro contains only one keystroke—F8. In Write, it looks like this:

```
<begdef><nokey><name>"Calculate"<vidon><f8><enddef>
```

I include the @Macro("Calculate") command in any field where I want to force a recalculation. I also use it to trigger a recalculation as soon as a selection is made from a list. This way, the user can see the results immediately, even before moving out of the field.

Erika Yoxall, Hammer Data Systems, 330-527-4018

Merge-Faxing . . . continued from page 1

Add a 10-character button field as shown in Figure 1. Name it **Send Fax**, format it *T, JC*, and set it to *Send Fax* at the Initial Values Spec. Then, program it like this:

```
< If Company = "" or Contact = "" or Fax = "" Then
  { @Msgbox("Company, Contact, and Fax",
    "must contain info for faxing.",
    "Please fix before proceeding."); Chome }
Else if @Askuser(
  "Send fax to", Contact + " at " + Company + "?", "" )
then { @Macro("Fax1"); Goto Company } Else Chome
```

To better understand what the program does, take a look at the *Fax1* macro it invokes:

```
<begdef><nokey><name>"Fax1"<vidon><f11><f7><dn><f12><capsf8>
fax1<enter><esc>yue4<enter>fax1.dbf<enter>y<altf8>fax1<enter><f10><altf8>
fax1<enter><f10><esc><esc>q<enddef>
```

When you click on the **Send Fax** button, the program makes sure none of the key fax-related fields are empty, then confirms that you want to send a fax to the current client. Though the `@Macro("Fax1")` command precedes the `Goto Company` command, Q&A executes the `Goto` first. Here are the steps the *Fax1* macro performs:

1. Copies the company to the Clipboard, presses F7 for the Retrieve Spec, pastes the company into the Company field, then saves the Spec to the name *Fax1*.
2. Creates a one-record dBASE export file with the selected company, contact, and fax number. (This becomes a WinFax *phonebook*, as you'll see.)
3. Returns to the Main menu, and presses "q" to switch to Q&A Write for Windows.

Before recording the macro, you'll need to create the dBASE export file (*FAX1.DBF*), and the saved Retrieve and Merge Specs (both named *Fax1*). Here's how:

1. Select File / Utilities / Export / dBASE IV. Enter the name of your database, then *FAX1.DBF* as the name of dBASE export file.
2. At the Retrieve Spec, type a company name that will




Figure 2. At the Write for Windows Retrieve Spec, the *Fax1* saved Retrieve (at the bottom of the screen) is the client you selected in Q&A for DOS.

A Marriage of Convenience

Write for Windows gives you a wealth of tools to create lavish documents that look the same onscreen as they do when printed or faxed. And when it comes to merging, Write for Windows has a lot over Write for DOS. You can include bitmap images (your company logo, for instance), and use If-Then-Else merge programming statements.

When you install Q&A for Windows, you can install just the word processor without the database, then link your copy of Q&A 5.0 to it so you can quickly switch between the two programs. This gives you all the benefits of a terrific Windows word processor that can merge data from your Q&A for DOS 3.0, 4.0, or 5.0 databases.

- retrieve *any record* with data in the company, contact, and fax fields. Press Shift-F8 and save the Spec to *Fax1*.
3. At the Merge Spec, type 1 in the company field, 2 in the contact field, and 3 in the fax field. Press Shift-F8 and save the Spec to *Fax1*, then F10 to run the export.

When recording the *Fax1* macro, have it overwrite the *Fax1* Retrieve Spec and *FAX1.DBF* export file.

Step 2—Create a WinFax dBASE Phonebook

In Write for Windows, when you print your merge-fax to the WinFax printer, WinFax will prompt for the recipient's name and fax number. Here's where the dBASE export file comes in. You turn it into a WinFax *phonebook* so you don't have to type anything.

Following are the steps to create a dBASE phonebook in WinFax Pro 7.x and 8.0. You'll only have to do this once. Thereafter, when you send a fax, your macro will update the phonebook by overwriting the dBASE file.

1. Start WinFax, and click on the Phonebook icon (the open book).
2. Click the *right* mouse button on *New Folder*, and select *New Phonebook* from the popup menu (See Figure 3.) Name the new phonebook *Fax1*.
3. Click on the *Advanced* button, and select *dBASE*. The *Select Phonebook File* dialog appears.
4. Navigate to your dBASE export file (*FAX1.DBF*), select it, click on the *Open* button, click on the *Map Fields* button, and the *Data Fields Assignment* dialog appears (See Figure 4.)
5. Link the fields by clicking and dragging each of the fields from the *Database Fields* list to the corresponding *Phonebook Properties* field. When you're done, click OK, then OK again, and WinFax will create your default phonebook.

Step 3—Install Q&A Write for Windows and apply the interoperability update

When you install Q&A for Windows, you can install just the Q&A Write word processor. Q&A 5.0 for DOS comes with an *interoperability* update that lets Q&A for Windows use Q&A 5.0 for DOS databases. Instructions on applying the update are in the README.WIN file that comes with Q&A 5.0, so I won't repeat them here. However, when you install *only* Write for Windows, you'll have to modify the update's Install.bat file. Here's how:

1. Open Install.bat in Write for DOS. (It should be in a QAWIN subdirectory under your main Q&A directory. If you can't find it, you probably didn't install the interoperability option, and you'll need to perform a selective reinstall from the Q&A 5.0 master disks.)

2. Type a *rem* command in front of these three lines:

```
rem if not exist %1\QAWIN.EXE      goto nofiles
rem if not exist %1\QAWINRES.DLL  goto nofiles
rem if not exist %1\QA.EXE       goto nofiles
```

3. Press Ctrl-F8 to resave, then run the update using the instructions in the README.WIN file. When you've done so, Write for Windows will let you select a Q&A 5.0 database as the merge file.

Step 4—Create a hot-link to Write for Windows

With Q&A 5.0 running in Windows 95, you can create a hot-link to Write for Windows using External Programs or Alternate Programs.

(If you're running Q&A 5.0 in Windows 3.1, the hot-link won't work—you'll have to toggle Q&A for DOS into a Window by pressing Alt-Enter, then start Q&A Write for Windows by double-clicking on its icon.)

To place the link on the External Programs menu, select Utilities / External Programs. In the Menu Options column, type *Q&A Write for Windows*. In the Command Line column, type *C:\QAWIN\QAWRITE.EXE*, substituting your own drive and path. Save the menu, and you can now switch to Write for Windows from anywhere in Q&A simply by pressing Alt-F9 and

choosing *Q&A Write for Windows*.

(If you plan to switch to Write for Windows from the fax recipient's record, you may have to set the database's Sharing Mode to *Allow*, and include in your Autoexec.bat file the command to load the DOS SHARE program.)

I added my hot-link to Q&A 5.0's Main menu. To do so, select Utilities / Set Alternate Programs. On an empty Alternate Program line, type the path and name to Write for Windows, such as *C:\QAWIN\QAWRITE.EXE*. On the Menu Selection line, type something like *Q&A Write Win*.

Step 5—Create the fax merge document

Figure 1 shows a sample Write for Windows fax-merge document. I won't cover how to use Write for Windows—that's a whole other subject. (If you've ever used a Windows word processor, you'll be on familiar ground.) I'll just mention a few things that apply to fax-merging.

The WinFax printer driver

When you install WinFax Pro, it adds its own printer driver to Windows printer list—but it's no ordinary driver. When you "print" a Write for Windows document to the WinFax "printer," it invokes WinFax. So, when creating your fax-merge document, start by pulling down the File menu, selecting *Print Setup*, and choosing *WinFax* from the printer list. You want Write to save the WinFax printer with the document.

A combination fax template

The combination fax-merge document in Figure 1 includes the cover page *and* the fax message. My faxes are usually just a few paragraphs, so I don't send a separate cover page. (If the fax is two or more pages, WinFax will send them all.) If you want a separate cover page, WinFax comes with a variety of styles you can use, or you can create one in Write for Windows.

Inserting merge fields

To insert a merge field, pull down the Insert menu, and select *Merge Field*. In the Merge Database dialog, the *List of Files Type* should be set to *Automatic*. Navigate to your Q&A 5.0 database, (see Figure 5), select it, click OK, and




Figure 3. WinFax's New Phonebook dialog. Type *Fax1* for the phonebook name, and select dBASE




Figure 4. WinFax's Data Field Assignment dialog. Here, you match the dBASE export file's fields with WinFax's own phonebook fields.

the fields list will appear. (See Figure 6.) Select a field, and click OK to insert it in the document. You can Insert additional merge fields using the icon on the toolbar.

When you've finished creating your fax-merge document, click on the File menu and select Save. In the Save Document dialog, enter a name for it, such as FAX1.QW, and click OK.

Load your fax-merge document at startup

To autoloading your fax-merge document when you switch to Write for Windows, select File / Preferences / File. Click on *Open Default Document* in the On Startup section. Click on the Select button, locate and select the merge document you just created, then click on OK.

Conclusion

There are a host of WinFax features I can't cover here. For one, you can send *attachments* with your fax—separate documents such as a price list or special announcement. The WinFax Pro *Users Guide* tells you how to create and send attachments.

You can sign your faxes electronically. Just have your signature scanned into a graphics file, then insert it in your fax-merge document.

By default, WinFax saves a copy of each fax it sends. To view one, start WinFax, click on the Send Log, and double-click on any fax to display it in the Fax Viewer. You can optionally print a hard-copy for filing.

With your hot-link to Write for Windows, you might want to use it for your mail-merge tasks as well. The only downside (other than a learning curve) is converting your existing Q&A for DOS merge documents. Though you can open your Write for DOS documents in Write for Windows, you'll have to replace any Write for DOS

Fax Area Codes

If most of your faxing is long-distance, you should have separate area code and fax number fields in your Q&A database (and thus in the dBASE export file that serves as the WinFax Phonebook). Otherwise, you'll have to type a "1" in front of any long distance number at WinFax's Send dialog box. If your fax numbers are in a 301-424-6755 format, and you'd like to separate the area code from the local number. Here's what to do:

1. Add a FaxAC (fax area code) field to your database.
2. Back up your database, then Mass Update it. At the Retrieve Spec, type /= in the original Fax field to select just the records that contain fax numbers. At the Update Spec, enter a program like this in the Fax field:

```
#1: If @Len(@Num(Fax)) = 10 Then {
    FaxAC = @Left(Fax, 3);
    Fax = @Del(Fax, 1, 3);
    If @Left(Fax, 1) = "-"
    Then Fax = @Del(Fax, 1, 1) }
```

3. Edit the Merge Spec for the dBASE export so it exports both the FaxAC field and the original Fax field, then relink your WinFax Phonebook to the dBASE file.

merge-field codes and programming expressions with their Write for Windows counterparts. The logic is similar enough, though, to make it pretty straightforward.

Tom Marcellus is author of *the PC World Q&A Bible*, and editor of *The Quick Answer*. Many of his Q&A databases are available from Marble Publications




Figure 5. Write for Windows' Merge Database dialog. Here, you select the Q&A database to use with your fax-merge document.




Figure 6. Write for Windows' Insert Merge Fields dialog shows the fields in the selected database. You can also add programmed fields here.



Stumped?

Send your Q&A questions to @Help, The Quick Answer, Marble Publications, Inc., PO Box 9034, Gaithersburg, MD 20898-9034 or fax to 301-424-1658. Please include your name, address, phone number, and your Q&A version number (and whether DOS or Windows) and a detailed description of the problem. We'll publish those questions we feel are of general reader interest; individual responses aren't possible.



Tutorial

Macros—The Missing Link

ERIKA YOXALL



Like that machine that answers your phone and plays your greeting message, Q&A macros can automate repetitive tasks for you. If you've shied away from macros in the past, here's your chance to groove-in from the ground up.



A client of mine with a pretty good grasp of Q&A programming was showing me a database she designed for her company. She demonstrated how users filled out the form, then printed it via the File / Print menu. When I mentioned that she could save keystrokes by using a macro to print the form, she said that sounded complicated and scary, and that she'd rather teach users to print the form manually. After I picked my jaw up off the floor, I walked her through the process of creating a print macro. She was amazed at how easy it was to create, and how smoothly the print process ran.

I checked around, and discovered that many users who are comfortable with other Q&A tools still shy away from macros. As a developer, I know that running a complex macro in front of a client for the first time can be a white-knuckle experience. But by the same token, macros are the cornerstone of any interface. Since Q&A just isn't Q&A without the automation macros provide, it's time for all you macrophobics to pinch your noses and jump in. Believe me, the water is nowhere near as cold as you might think.

What exactly is a macro, anyway?

Probably the most intimidating thing about a "macro" is the name. It comes from the Greek *makro*, and means large, as in *macronucleus*, or inclusive, as in *macroinstruction*. On a computer, it's just a single instruction that results in a sequence of actions. If I tell a client that the computer can perform some task automatically, I see excitement. If I suggest that we "write a macro" to perform the task, I see eyes glaze over.

Whatever you call it, a macro works on the same basic principle as an answering machine. You record your greeting one time, and the machine plays back the same greeting whenever the phone rings.

When you record a macro, you perform the task manually (press the keys) one time, then tell Q&A to play back those same keystrokes whenever you want to perform the task. Any repetitive task you can perform manually, you can record in a macro for Q&A to perform.

Testing the waters

Before you start experimenting with creating macros, you should build a small isolated environment where you can get your feet wet without having to worry about data the macro might affect. Follow these steps:

1. Check with your database administrator regarding your current macro file. A backup copy of it should be made just in case. Q&A's default macro file is named QAMACRO.ASC, and it's usually stored in the main Q&A directory.
2. Create a subdirectory for your test files, such as C:\QA\MACTEST.
3. From Q&A's Main menu, select Write / Type/Edit, and save a blank document to your new directory. Press Ctrl-F8 to save it as an ASCII file, and name it C:\QA\MACTEST\QAMACRO.ASC. This way, it will behave like Q&A's default macro file.
4. Design a simple "answering machine" database named ANSMACH.DTF in your test subdirectory, and customize it as follows:

Field Name	Format / Attributes
MessageDate	Date (D), Read-only
Name	Text (T)
Message	Text (T), Multiline field (five lines)
LeaveMessageButton	T,J,C, 15 characters wide. Define using the < and > characters, and make the field Light Yellow on Black at the Palette Spec.

Type a "greeting" right into the form design as shown in Figure 1, then save it.



Figure 1. Your ANSMACH.DTF form design should look this when finished.

5. Go to the Program Spec for ANSMACH.DTF. Press F6 to expand the Message Date field, and type the following program:

```
#10: MessageDate = @Date;  
LeaveMessageButton = "Leave Message"
```

6. Close the field, move to the Message field, and type this program:

```
> Goto Name
```

Press F8, and type 10 on the On Entry field ID line. Save the Program Spec

Now that you have a safe place to work, you're ready to record your first macro. The focus of this exercise is working with macros. The practice database isn't designed to do anything complex or even interesting.

Get your feet wet

Let's create a simple macro. This one will open ANSMACH.DTF to receive a new message. (To do this manually, you'd choose File / Add data, type ANSMACH at the filename prompt, and press Enter.) Start by activating the new macro file you created earlier:

1. Go to the Q&A Main menu, and press Shift-F2 to display the Macro Menu.
2. Select Get Macros, type the path to your test macro file, such as C:\QA\MACTEST\QAMACRO.ASC, and press Enter.

To create then save your macro, follow these steps:

3. At the Q&A Main menu, press Shift-F2 to display the Macro menu, and choose Define Macro.
4. A message will appear at the bottom of the screen prompting you to assign a macro identifier. Press Alt-L. (Hold down the Alt key, and press the lowercase L.)

A flashing rectangle will appear in the lower right corner of the screen indicating the macro recorder is running. Any keys you now press will be recorded.

5. Select File from the Q&A Main menu by pressing F. (If your menus aren't set for Automatic Execution, you'll have to press F, then press Enter.)
6. Choose Add data from the File Menu by pressing A. (Press A then press Enter if your menus aren't set for Automatic Execution.)

7. At the filename prompt, press Home, type the path to ANSMACH.DTF (C:\QA\MACTEST\ANSMACH, for example), then press Enter. A blank ANSMACH record should appear.

8. Press Shift-F2 to end recording. The Macro Name field will show the Alt-L identifier you assigned to the macro. Type a meaningful name over it, such as *Leave a Message*.

9. Leave the Show Screen line set to Yes, leave the End With Menu line empty, and press F10.

10. Q&A will ask you if you want to save this macro. Make sure that the macro filename Q&A proposes is your test macro file, and press Enter.

11. Reload your test macro file by repeating steps 1 and 2 under "Start by activating the new macro file you created earlier." This lets Q&A reread the macro file without having to exit and restart Q&A. This safety measure can prevent damage to macros.

To test your new macro, escape from the record, and return to the Main menu. Press Alt-L, and Q&A should display a new ANSMACH record. Congratulations! You just created a macro! A task that required up to 24 keystrokes now takes just two. You can now have your database administrator reactivate the usual macro file.

Some notes on macro-making

- Whenever possible, avoid using the Arrow keys to make selections. The order of selections on lists and even menus can change.
- When Q&A prompts for a filename, always press Home (or Shift-F4 to clear the prompt line), then enter the entire path and filename. Q&A always proposes the path and filename of the last opened file, and these might not always be correct.
- If you make a mistake while recording a macro, press Shift-F2, then Esc to clear the macro and start over.

In my next installment on macros, I'll go into more detail on the selections on the Macro menu. We'll also look at different macro elements, and I'll show you how to edit a macro in the Write word processor.

Erika Yoxall owns Hammer Data Systems in Garrettsville, Ohio, specializing in Q&A and Microsoft Access. 330-527-4018, emy103@worldnet.att.net.



Erika Yoxall
Tom Marcellus



MEET AND HEAR THESE DASHING Q&A EXPERTS OCTOBER 24-26 IN SAVANNAH, GEORGIA

Gordon Meigs
Bill Halpern



Reserve your place now at the **National Q&A User Group's 7th Annual Bash and Quick Answer Masters Seminar**. Watch live Q&A demos on the Big Screen. Bring your nagging Q&A questions. Hobnob with fellow Q&A enthusiasts. For details on the Bash, visit the group's Web site at <http://qaug.com>, or write to Gale Platt, 9233 SW 8th St. #423, Boca Raton, FL 33428. For info on the Masters Seminar, call **800-780-5474**.

Spend a fun-filled, informative weekend with fellow Q&A diehards from around the world. See y'all there!

Selection Lists Make Data Entry a Breeze



JEFF NITKA

In multi-line-item databases for invoices, purchase orders, and the like, you can't beat point-and-shoot data entry and automatic sorting.



NOTHING brings speed and accuracy to data entry like pop-up selection lists. But pop-up lists can do more than fill fields—they can display the right user-selectable *options* at key points. By teaming up *data* selection with *option* selection, you can make the data entry process a no-brainer.

I developed an invoice database that requires virtually no typing. All user interaction—including adding the line-items *and* choosing various options at key points—is taken care of by pop-up lists. But the database does even more:

- You can add invoice items in any order, then sort them by product name or description. Any duplicate line-items are caught as they're added.
- Any line-item can be deleted at any point, and the remaining ones will be moved up to fill the gap.
- Calculated and lookup fields are read-only. The user can't manually change the invoice total, an item's price or description, and so forth.

I'll show you a somewhat scaled-down version of this database. Using the programming, you can add any or all of these features to your own line-item databases. (To obtain the database, see "Resource" on page 10.)

Using the database

Here, in a nutshell, is how INVOICE.DTF works. When adding a new invoice, you can let the program assign the next invoice number in sequence, or you can enter the invoice number manually to continue an invoice.

Next, you choose the customer from a list, and the billing information fields are filled in by lookups.

You then start adding the line-items. You select the first item from a list, and lookups return the product name, description, and unit price. You enter the quantity, and the program calculates the extended amount.

The next list prompts you to add another item, delete an item, or indicate that you're finished adding items. The option to add an item appears only if there's room on the invoice. When you delete an item, any items below it are

automatically moved up.

At any time, you can sort the items by product name or description, save the invoice and start another one, save the invoice and exit, or return to the original options (add, delete, sort, and so forth).

As you can see, pop-up lists aren't used just to fill the invoice fields. They also prompt you with *appropriate* options at key points during the data entry process.

Database design

Figure 1 shows the sample Invoice database in the Design Spec, but with the field names added for reference. Here's how the fields are formatted and customized:

Date field: InvDate

Text fields: Cust Name, Address1, Address2, Location, Start, Prod1 through Prod10, Desc1 through Desc10, List, Search, InvNo

Number fields: Ptr, Count, Entry, Qty1 through Qty10, Total Qty

Money fields: Price1 through Price10, ExtPr1 through ExtPr10, Total Ext

Yes/No field: Changes

Read Only fields: InvDate, Cust Name, Address1, Address2, Location, Total Qty, Total Ext, Ptr, Count, Start, List, Changes, Prod1 through Prod10, Desc1 through Desc10, Qty1 through Qty10, Price1 through Price10, ExtPr1 through ExtPr10

Invisible fields: Start, List, Search, Ptr, Count, Changes

Most of the fields are read-only. This prevents users from changing any data that's calculated or retrieved from an external database. For example, users can add or remove entire line-items, but they can't change a price, description, or extended amount.

Invoice No < InvNo >		Bill To < Cust Name > < Address1 >		< Start >
Invoice Date < InvDate >		< Address2 > < Location >		< Entry >
Item No	Description	Price	Qty	Ext Price
1	<Prod1 > <Desc1 >	> Price1	Qty1	ExtPr1
2	<Prod2 > <Desc2 >	> Price2	Qty2	ExtPr2
3	<Prod3 > <Desc3 >	> Price3	Qty3	ExtPr3
4	<Prod4 > <Desc4 >	> Price4	Qty4	ExtPr4
5	<Prod5 > <Desc5 >	> Price5	Qty5	ExtPr5
6	<Prod6 > <Desc6 >	> Price6	Qty6	ExtPr6
7	<Prod7 > <Desc7 >	> Price7	Qty7	ExtPr7
8	<Prod8 > <Desc8 >	> Price8	Qty8	ExtPr8
9	<Prod9 > <Desc9 >	> Price9	Qty9	ExtPr9
10	<Prod10 > <Desc10 >	> Price10	Qty10	ExtPr10
Grand Totals		Quantity < Total Qty >		Extended Price < Total Ext >
Count < > List < > Changes < > Ptr < > Search < >				

Figure 1. The sample Invoice database at the Design Spec. Field names are added for reference.

Invisible (hidden) fields are those you don't need to see during data entry. To make a field invisible, use the Palette Spec to make its text and background color the same as the form's color.

Programming

The following two Navigation Spec programs are required. Listing 1 shows the complete Program Spec:

```

InvNo: < if @Add then
@Msg("Enter invoice number if continuing an " +
"invoice. Leave blank if new invoice.")

Entry: < @Msg("Enter the quantity to ship for Item #" +
@Str(Count+1) + ".")

```

Listing 1. Program Spec for the invoice database.

```

InvNo
> if @Add then
if InvNo = "" then
{ InvNo = @Str(@Tn(@Xlr(@fn,9999999,"InvNo",
"InvNo")+1);
InvNo = @Txt(5-@len(InvNo),"0") + InvNo }
else
if @Xlu(@fn,InvNo,"InvNo","InvNo") <> ""
then if @AskUser(InvNo + " already exists. Do",
"you wish to add to this invoice?","",")
then Xlu(@fn,InvNo,"InvNo","InvDate",InvDate,
"Cust Name",Cust Name,
"Address1",Address1,
"Address2",Address2,
"Location",Location)
else Goto InvNo
else
if @AskUser(InvNo + " not currently in file.",
"Is this the Invoice Number you want?","",")
then Goto Start
else Goto InvNo
else Goto Search; Goto Search

Start:
< if @Add then
{ if Cust Name = "" then
{ Cust Name=@XUsR("Contacts","Name","Name","",");
if Cust Name = ""
then Goto Search
else { Xlu("Contacts",Cust Name,"Name",
"Address",Address1,
"City/St/Zip",Location) };
Clear(Prod1..Prod10,Desc1..Desc10,List,
Ptr,Qty1..Qty10,Price1..Price10,Search,
Start,Count,ExtPr1..ExtPr10,Entry);
Goto Search } }

Entry:
> if Entry <= 0 then
if @AskUser("Entry not allowed! Do you wish to",
"try again? (Answering NO will ",
"delete your current entry)")
then { Entry = ""; Goto Entry }
else { Clear( @("Prod"+@Str(Ptr)),
@("Desc"+@Str(Ptr)),
@("Price"+@Str(Ptr)),
@("Qty"+@Str(Ptr)),
@("ExtPr"+@Str(Ptr)) );
Goto Search }
else { List = "*" + @("Prod"+@Str(Ptr)) + "*" + List;
Count = Count + 1;
@("Qty"+@Str(Ptr)) = Entry;
@("ExtPr"+@Str(Ptr)) = Entry *
@("Price"+@Str(Ptr));
Total Qty = @Sum(Qty1..Qty10);
Total Ext = @Sum(ExtPr1..ExtPr10);
Entry = ""; Goto Search }

Search:
< Ptr = Count + 1;
if Cust Name = "" then

```

```

{ Search = @Us("1-Choose Customer,2-Quit");
if @Lt(Search,1) = "1" then Goto Start }
else
if Count > 0 and Count < 10
then { @Msg("You can add or delete " +
"items for this invoice.");
Search = @Us("1-Add Line-Item," +
"2-Delete Line-Item," +
"3-Sort Line-Items," +
"4-Finished,5-Quit") }
else
if Count = 0 then
{ @Msg("You can choose up to 10 " +
"items for this invoice.");
Search = @Us("1-Add Line-Item,2-Quit") }
else { @Msg("No more items can be entered.");
Search = @Us("1-Delete Line-Item," +
"2-Sort Line-Items," +
"3-Finished,4-Quit") };

if Search = "" then Goto Search else
{ if @in(Search," ") > 0
then Search= @mid(Search,3,@in(Search," ") -2)
else Search= @rt( Search, @len(Search) - 2);
if Search = "Add"
then { @Msg("Choose an item, " +
"then press ENTER.");
Search = @XUs("Products","Prod Name");
if Search = "Add" or Search = ""
then Goto Search else
Xlu("Products",Search,"Prod Name",
"Prod Name",@("Prod"+@Str(Ptr)),
"Description",@("Desc"+@Str(Ptr)),
"Price", @("Price"+@Str(Ptr)));
if @In(List,"*" + @("Prod"+@Str(Ptr)) + "*" > 0)
then { @MsgBox("You've already entered " +
@("Prod"+@Str(Ptr)) + ". ",
"Please make a different selection.", "");
Clear(@("Prod"+@Str(Ptr)),
@("Desc"+@Str(Ptr)),
@("Price"+@Str(Ptr)));
Goto Search }
else Goto Entry }
else
if Search = "Delete"
then { @Msg("Choose an item to delete (press" +
" ESC if you don't want to delete).");
Search = @Us("01 - Item 1" +
@txt(Prod2<>""",",02 - Item 2") +
@txt(Prod3<>""",",03 - Item 3") +
@txt(Prod4<>""",",04 - Item 4") +
@txt(Prod5<>""",",05 - Item 5") +
@txt(Prod6<>""",",06 - Item 6") +
@txt(Prod7<>""",",07 - Item 7") +
@txt(Prod8<>""",",08 - Item 8") +
@txt(Prod9<>""",",09 - Item 9") +
@txt(Prod10<>""",",10 - Item 10") );
if Search = "" then Goto Search else
{ Ptr = @Tn( @Lt(Search,2) );
List = @Replace(List,"*" +
@("Prod"+@Str(Ptr)) + "*" + "");
Clear( @("Prod"+@Str(Ptr)),
@("Price"+@Str(Ptr)),@("Desc"+@Str(Ptr)),
@("Qty"+@Str(Ptr)),@("ExtPr"+@Str(Ptr)));
Count = Count - 1;
GoSub Ptr; Goto Search } }
else
if Search = "Finished"
then { Search = @Lt(@Us("1-Save/Do Another," +
"2-Save and Exit,3-Make Changes"),1);
if Search = "" or Search = 3
then Goto Search else
if Search = 1 then @Save else @Exit }
else
if Search = "Sort"
then if Count > 1
then { Search = @Us("1-By Product," +
"2-By Description,3-Cancel Sort");
Search = @Lt(Search,1);

```

```

        if Search = "" or Search = "3"
        then Goto Search
    else Start= @Txt(Search="1","Prod")+
        @Txt(Search="2","Desc");
        Ptr = 1; Changes = 0;
        GoSub Count; Goto Search }
    else Goto Search
else
if @AskUser("Are you sure you want",
    "to cancel this operation?","")
then @Macro("Exit") else Goto Search }

Ptr:
< if Ptr = 10 then
{ Total Qty = @Sum(Qty1..Qty10);
  Total Ext = @Sum(ExtPr1..ExtPr10); Return }
else
if @("Prod"+@Str(Ptr+1)) = "" then
{ Total Qty = @Sum(Qty1..Qty10);
  Total Ext = @Sum(ExtPr1..ExtPr10); Return }
else { @("Prod"+@Str(Ptr))=@("Prod"+@Str(Ptr+1));
  @("Desc"+@Str(Ptr))=@("Desc"+@Str(Ptr+1));
  @("Price"+@Str(Ptr))=@("Price"+@Str(Ptr+1));
  @("ExtPr"+@Str(Ptr))=@("ExtPr"+@Str(Ptr+1));
  @("Qty"+@Str(Ptr))=@("Qty"+@Str(Ptr+1));
  Clear( @("Prod"+@Str(Ptr+1)),
    @("Desc"+@Str(Ptr+1)),
    @("Price"+@Str(Ptr+1)),
    @("ExtPr"+@Str(Ptr+1)),
    @("Qty"+@Str(Ptr+1)) );

    Ptr = Ptr + 1; Goto Ptr }

Count:
< if Start <> "" then
{ if @("Start"+@Str(Ptr)) > @("Start"+@Str(Ptr+1))
  then { Changes = 1;

    Search = @("Prod"+@Str(Ptr));
    @("Prod"+@Str(Ptr)) = @("Prod"+@Str(Ptr+1));
    @("Prod"+@Str(Ptr+1)) = Search;

    Search = @("Desc"+@Str(Ptr));
    @("Desc"+@Str(Ptr)) = @("Desc"+@Str(Ptr+1));
    @("Desc"+@Str(Ptr+1)) = Search;

    Search = @Str( @("Price"+@Str(Ptr)) );
    @("Price"+@Str(Ptr)) = @("Price"+@Str(Ptr+1));
    @("Price"+@Str(Ptr+1)) = @Tm(Search);

    Search = @Str( @("Qty"+@Str(Ptr)) );
    @("Qty"+@Str(Ptr)) = @("Qty"+@Str(Ptr+1));
    @("Qty"+@Str(Ptr+1)) = @Tm(Search);

    Search = @Str( @("ExtPr"+@Str(Ptr)) );
    @("ExtPr"+@Str(Ptr)) = @("ExtPr"+@Str(Ptr+1));
    @("ExtPr"+@Str(Ptr+1)) = @Tm(Search); };

    if Ptr < Count - 1
    then Ptr = Ptr + 1 else
    if Changes
    then { Changes = 0; Ptr = 1 }
    else { @Msg("Sort Complete"); Start = "" };

    Goto Count }
else { Ptr = Count + 1; Return }

```

What the programming does

In this invoice, the cursor never enters a line-item field—all the action takes place in the special fields in Listing 1. This makes it easier to control data entry flow, and makes the program as a whole more flexible. If you redesigned the invoice with more or fewer line-items, it would be easy to modify the programming to accommodate them.

The **InvNo** field prompts for an invoice number (in case an invoice is being continued), or assigns the next highest invoice number if the field is left empty.

The **Start** field displays a selection list of customers,

and retrieves the billing information. It then clears the other fields and passes control to the Search field.

The **Search** field contains the workhorse program. All list manipulation is controlled by it. The various pop-up lists offer different choices depending on how many line-items are in the record. For example, if the line-items are all empty, the sort and delete options aren't offered. If all the line-items are filled, the add option isn't offered.

The **Ptr** (pointer) field contains the next available line-item number. It's also used for sorting and to adjust for line-item deletions. It works in conjunction with the **Count** field, which contains a modified version of the bubble-sorting routine in the *Q&A Application Programming Tools Manual*. If the invoice contains five items, for example, and you delete the third one, the subsequent items are moved up a line.

The **List** field tracks the line-items to prevent duplicate entries. **Changes** is used by the sorting routine to check if any line-items have changed position.

Conclusion

Besides saving time, creating invoices using selection lists reduces errors. The sample database obviously doesn't include all the fields you'd need in an actual invoice, but it's a good starting point in developing a line-item-type database that facilitates (and controls) data entry via selection lists.

If your database is designed to accommodate fewer or more than 10 line-items, you'll need to make a few programming changes. In the **Ptr** field's first line (If **Ptr** = 10 Then...), change the "10" to the number of line-items the invoice is designed to accommodate. Near the beginning of the **Search** field's program (if **Count** > 0 and **Count** < 10...), change the 10 to the same number of line-items. Finally, in the following **Search** field program, add or delete the number of instances of the **@Text** function:

```

Search = @Us("01 - Item 1" +
    @txt(Prod2<>"",",",02 - Item 2") +
    @txt(Prod3<>"",",",03 - Item 3") +
    @txt(Prod4<>"",",",04 - Item 4") + ...etc.

```

Jeff Nitka works for a chemical manufacturer and develops Q&A applications part-time for Epoch Software, 908-874-3989. Jeff is the author of the Program Evaluator (a Q&A program debugging utility), SurfDrive (a Q&A disk drive reader), and FaxMan (a Q&A faxing database), all available from Marble Publications.

Resource

The invoice database featured in this article (the programming and design are unlocked, and can be modified and copied), plus the two lookup databases, are available postpaid for \$25 from Marble Publications, PO Box 9034, Gaithersburg, MD 20898. Phone chargecard orders to 800-780-5474 or 301-424-1658, or fax to 301-424-1658. Checks, Visa, and Mastercard okay. Ask for disk **JN-0797**.



EDITED BY T.J. SHUFLIN

Combining Date Components



I use Q&A 5.0, and I have my dates broken down into three fields—Month, Day, and Year. I'd like to combine them into an Order Date field, but I'm stumped on how to do it?

Jerry Bailey, Centerville, Ohio

You didn't say if the original three fields are formatted for text or numbers, so I'll show the programming for both. (I'll assume Order Date is a Date field.) If the original fields are all formatted for text (T), use this program, substituting your own field ID numbers:

```
Month: #10
Day: #20
Year: #30
Order Date:
<#40: #40 = @Todate( #10 + "/" + #20 + "/" + #30 )
```

If the original fields are all formatted for Numbers (N), use this Order Date program instead:

```
< #40: #40 = @Todate( @Str(#10) + "/" + @Str(#20) + "/" +
@Str(#30) )
```

As you may know, several of the date formats in Q&A 5.0 are buggy. Three tips in past issues offer suggestions on what to do about them. See March 1996, p. 8 (two tips), and March 1997, p. 6. Here's another date field oddity: if you inadvertently enter a month greater than 12 and a day that's 12 or less, Q&A will swap the month and day! (This can happen even when the date value is constructed by programming, as above.) If the month and day are both out of range, Q&A will warn you that it can't format the value as a date.

Fonted Reports Print Bare Bones



I use Q&A 5.0 with an HP LaserJet 4Si printer. I've designed reports that include various fonts and font sizes. When I print one without making changes, all my fonts print fine. In some cases, though, if I temporarily change a report's Retrieve Spec, the entire report prints in the default printer font. What's happening, and how can I avoid this?

David J. Mount, New York

Tom Marcellus provided his thoughts on the problem, and I was able to duplicate it. So, here's our collaborative answer. The loss of fonts in the report output is caused by

pressing F6 to open the Field Editor. If you answer Yes to Temporary Changes after selecting Report / Print a Report, then press F6 at any report screen, Q&A will ignore your font assignments when printing the report.

In contrast, when you use Design / Redesign a Report, pressing F6 doesn't open the Field Editor, but expands the field at the bottom of the screen. So changes you make to a report here won't cause the loss of fonts.

To avoid opening the Field Editor, one of the following workarounds might prove helpful:

- If a temporary parameter is too long for the field, type it in a larger field, use F11 then F12 to copy then paste it into the correct field, then delete it from the other field.
- Redesign the database, making the field(s) you normally modify during Temporary Changes large enough to accommodate the changes.
- Use Design / Redesign a Report to make changes.
- Select Rename/Delete/Copy at the Report menu. Copy the report to a new name, and make changes to the copy in Design / Redesign mode. Delete the copied report once it has been printed.

Special Date Calculation in Report Header



I need a report header to print the date of the last day of the previous month in a header line like this:

Printed on (current date) for period ending (date of last day of previous month)

David J. Mount, New York.

Once again, Tom Marcellus provided the nuts and bolts for this answer. (See his editorial in the May 1997 issue.) Create two derived columns like these:

```
Heading:
Formula No. 1: (@Date - @Dom(@Date))
Column/Sort Spec: 4,I
```

```
Heading:
Formula No. 2: @Month$(#4) + " " + @Str(@Dom(#4)) + ", "
+ @Str(@Year(#4))
Column/Sort Spec: 5,I
```

Then, in your report header, type this line:

Printed on @Date(11) for period ending *PG {?2}*

Here, date format #11 is specified, so if the current date is July 15, 1997, the header should print like this:

Printed on July 15, 1997 for period ending June 30, 1997

The *PG {?2}* header command prints the results of the second derived column's calculation. (Be sure to include

the space in front of the first French brace.) I used two derived columns to make the solution easier to understand, but you can do it with one. Just replace each occurrence of #4 in Formula No. 2 with Formula No. 1. Then, if this is now the first derived column, change the 2 to 1 in the header program, so it reads *PG {?1}*.

What prints in the header is the result of the derived column's formula for the *first* record in the report. (Since the formula is based on @Date, it's the same for all records anyway.) To base the derived column formula on a date field in the database (or a date column in the report), simply replace all occurrences of @Date with the date field or date column number.

Rearranging Derived Columns



I've upgraded to Q&A 5.0 for DOS, and am learning how to use Derived Columns. Sometimes I have to add new derived columns ahead of those I've already created. It's a real pain to move the existing ones down to make room.

Mary Messina, Boyce, Louisiana

You can add a derived column with a lower column number *after* columns with higher numbers. However, it's still a good practice to keep derived columns in ascending column order as it makes the report design easier to follow, and helps ensure any calculations are performed in the correct sequence.

You've upgraded to Q&A 5.0, so you can use the Clipboard. To move more than one derived column, move the last one a line at a time to the new location—which might be several screens down. This way, you can overwrite information you've already copied to a screen even further down. Simply press F11 on any line to copy it, move to the new line, and press F12 to paste it.

Before redesigning a report, it's a good practice to copy it to a new name, then work with the copy. This way, if you make an error, you'll still have the original report design to fall back on.

T.J. Shufflin is a forensic chemist and crime lab director in Alexandria, Louisiana. He served as the National Q&A User Group's first president in 1991-92, and remains active by maintaining the group's Web site at <http://www.qaug.com>.



Privatize Files



Got a file or two you'd like to hide from prying eyes? Here's an easy way to make a file invisible in Q&A, then

make it visible again to work with it. Design a password-protected database that contains two fields—**Filespec** and **ReturnCode**—then program them along these lines:

```
Filespec:  
< Userselect("Hide,Unhide", ReturnCode);  
  If @Instr(ReturnCode, "hide") Then  
    @Msg("Enter the path and name of file to " + ReturnCode)
```

```
ReturnCode:
```

```
< If ReturnCode = "Hide" Then {  
  ReturnCode = @Shell("Attrib " + Filespec + " +h");  
  @Msg(Filespec + " is now hidden");  
  ReturnCode = "Hidden" };  
  
  If ReturnCode = "Unhide" Then {  
  ReturnCode = @Shell("Attrib " + Filespec + " -h");  
  @Msg(Filespec + " is now unhidden");  
  ReturnCode = "Unhidden" }
```

The program uses the DOS ATTRIB (attribute) command to hide or unhide the specified file. You can use *+r* and *-r* instead of the *+h* and *-h* to make and unmake a file read-only.

Tom Marcellus



PO Box 9034
Gaithersburg, MD 20898-9034

Periodicals
Postage
PAID
at Rockville, MD